ETH ***EIDGENÖSSISCHE TECHNISCHE HOCHSCHULE ZÜRICH***

Analysis

mit dem Computer-Algebra-System des TI-92

Teil 2:
Ableitung

Beat Eicke und Edmund Holzherr
11. November 1997

Eidgenössische Technische Hochschule
CH – 8092 Zürich

INHALT

Seite

Hinweise

TI-92 - Notationen

Wir haben folgende Notationen bei Berechnungen mit dem TI-92 verwendet:

$\rightarrow$ oder auch –> bedeutet, dass die Taste STO ▸ zu drücken ist

$\Rightarrow$ Nach einem Doppelpfeil folgt stets ein Resultat des TI-92.

Aufgaben

Wir haben keine Sammlung mit neuen oder gar revolutionären Aufgaben herausgeben wollen. Trotzdem haben wir ab und zu einige Vorschläge eingetreut, um zu zeigen, welche Aufgabentypen (nach wie vor) verwendet werden könnten, weil sie durch den Einsatz des CAS nicht völlig trivialisiert werden, sondern durchaus noch einige mathematische Kenntnisse voraussetzen.

2. Ableitung

2.1 Einführung

PROBLEM: Velo - Blitzstart

Welche Geschwindigkeiten erreicht man mit einem Rennvelo während der Startphase? Wie steht es mit der Beschleunigung ?

MODELLIERUNG (Experiment mit konstanter Übersetzung)
Bei einem Test wurden in bestimmten Abständen die Zeiten gemessen.

Weg in m: s	0	5	10	20	30	50	75	100
Zeit in sec: t	0	2.78	3.79	5.15	6.17	7.74	9.27	10.54

Mit dem Computer (TI-92 als Black-Box) bestimmen wir zuerst näherungsweise die Funktion[1] $s : t \mapsto s(t)$
Dazu geben wir die Daten als Listen ein.

{ 0, 2.78, 3.79, 5.15, 6.17, 7.74, 9.27, 10.54} -> t
{ 0, 5, 10, 20, 30, 50, 75, 100} -> s
Newplot 1 , 1, t , s

◆ [WINDOW]
xmin=-1 xmax=11 ymin=-1 ymax=100

◆ [GRAPH]



Das Punktediagramm legt den Ansatz des Modell $s = a\,t^b$ nahe. Für dieses Modell ist $t = 0$ beim TI-92 nicht erlaubt. Es ist zum vornherein $s(0) = 0$ definiert. Wir müssen die Datenlisten korrigieren.

{ 2.78, 3.79, 5.15, 6.17, 7.74, 9.27, 10.54} -> t
{ 5, 10, 20, 30, 50, 75, 100} -> s

PowerReg t , s
ShowStat ⇒
RegEq (x) -> y1(x)
NewPlot 1 , 1, t , s
◆ [GRAPH] (Punktediagramm mit dem Graph)



Für die weiteren Untersuchungen verwenden wir das leicht vereinfachte Modell :

$$s(t) = \frac{1}{2}\, t^{\frac{9}{4}} \qquad (1)$$



[1] Beim Versuch ist t die abhängige und s die unabhängige Variable; die Vertauschung ist aber nicht wesentlich

Die mittlere Geschwindigkeit in einem sehr kleinen Zeitintervall [t , t + Δt] ist die mittlere Änderungsrate des Weges s :



DelVar t , s
tt^(9 / 4) / 2 -> s (tt) ⇒ Done
(s (t1+Δt1) - s (t1)) / Δt1 -> vm (t1 , Δt1)

vm (t , Δt) ⇒ $\dfrac{(t+\Delta t)^{9/4} - t^{9/4}}{2\,\Delta t}$

Bemerkungen
- Eingabe von Δ: [G] [↑] [D]
- Bei der Definition und dem Aufruf einer Funktion darf man nicht die gleiche Variable verwenden.

INTERPRETATION
Die mittlere Geschwindigkeit vm (t, Δt) ist die STEIGUNG DER SEKANTE IM INTERVALL [t , t + Δt].

DARSTELLUNGEN DER MITTLEREN ÄNDERUNGSRATE **vm(t , Δt)** :

1. **Tabelle**
   [MODE] Graph -> 1: FUNCTION [ENTER] [ENTER]
   vm (2,x) -> y1(x)

   [Y=] y2(x)=vm (4,x)
   y3(x)=vm (6,x)
   y4(x)=vm (8,x)

   [F] Axes ......OFF-> 2: AXES

   [TblSet] tblStart: 0. 35 Δtbl :- 0.05

   [TABLE]

   [TblSet] tblStart: 0. 07 Δtbl :- 0.01

   [TABLE] usw.



x	y1	y2	y3	y4
.35	2.9725	6.7145	10.951	15.552
.3	2.9296	6.6641	10.896	15.492
.25	2.8869	6.6138	10.84	15.433
.2	2.8443	6.5637	10.785	15.373
.15	2.8019	6.5136	10.73	15.314
.1	2.7597	6.4636	10.675	15.255
.05	2.7176	6.4137	10.619	15.195
0.	undef	undef	undef	undef

x=.35
MAIN RAD AUTO FUNC



x	y1	y2	y3	y4
.07	2.7344	6.4337	10.641	15.219
.06	2.726	6.4237	10.63	15.207
.05	2.7176	6.4137	10.619	15.195
.04	2.7092	6.4038	10.608	15.183
.03	2.7008	6.3938	10.597	15.172
.02	2.6925	6.3839	10.586	15.16
.01	2.6841	6.3739	10.575	15.148
0.	undef	undef	undef	undef

x=.07
MAIN RAD AUTO FUNC

   INTERPRETATION
   Die Resultate aus der Tabelle lassen vermuten, dass für Δt -> 0 die Grenzwerte von vm (t, Δt) mit t= 2, 4, 6 , 8 existieren.

2. **3D-Graph**
   [MODE] Graph ....FUNCTION-> 5: 3D [ENTER] [ENTER]
   vm (x,y) -> z1(x,y)



   [Y=]

   [F] Axes ....OFF-> 2: AXES

   [WINDOW]

   eyeθ°= -165 eyeϕ°=50
   xmin=0 xmax=12 xgrid=1 (= t)
   ymin=0 ymax=1. 5 ygrid=25 (= Δt)
   zmin=0 zmax=20 (= vm(t, Δt))
   [F2] 6: ZoomStd

   [GRAPH]

   INTERPRETATION:
   - der Graph von vm (t, Δt) ist eine Fläche
   - dass der Graph von vm (t, Δt) für Δt = 0 existiert, kann man vermuten, aber nicht sicher erkennen.

ZOOM-Operation I

Die mittlere Geschwindigkeit (mittlere Änderungsrate des Weges) in immer kleineren Intervallen:
Bezeichnung für den TI-92: y1 (x) = s (t)

Wir legen die Sekanten durch den Punkt P (2 , y1(2)) und einen 2. Punkt, der immer näher an P heranrückt.

Gerade durch zwei Punkte P (2 , y1 (2)) und Q (5 , y1 (5)): $y2(x) = \dfrac{y1(5) - y1(2)}{5-2}(x-2) + y1(2)$

 [Y=]

y1 (x)=x^(9/4) /2
y2 (x)=(y1 (5)-y1 (2))/3*(x-2)+y1 (2) F6 2: Dot
y3 (x)=(y1 (4)-y1 (2))/2*(x-2)+y1 (2) F6 2: Dot
y4 (x)=(y1 (3)-y1 (2))/1*(x-2)+y1 (2) F6 2: Dot
y5 (x)=(y1 (2.5)-y1 (2))/0.5*(x-2)+y1 (2) F6 2: Dot

 [WINDOW]
xmin=0 xmax=10 ymin=0 ymax=20

 [GRAPH]





[F2] Zoom 1: ZoomBox
↖ xc:1.01
↘ xc:3.86

[F5] Math [A]: Tangent 2
⇒ y = 2.67572 x - 2.97302 (Tangente)

ZOOM-Operation II



 [Y=] y2 ,..., y5 löschen

 [WINDOW]
xmin=0 xmax=10 ymin=0 ymax=20

 [GRAPH]

[F2] Zoom 1: ZoomBox ↖ xc:1.01
↘ xc:4.04

[F5] Math [A]: Tangent 2
⇒ y = 2.67572 x - 2.97302 (Tangente)



[F2] Zoom 1: ZoomBox ↖ xc:1.51
↘ xc:2.34

[F5] Math [A]: Tangent 2
⇒ y = 2.67572 x - 2.97302 (Tangente)

Feststellung:
In einer kleinen Umgebung von P(2, y1 (2)) ist die Tangente eine gute Näherung für den Graphen von y1 (x).

Momentane Geschwindigkeit :
Die momentane Geschwindigkeit v(t) zum Zeitpunkt t ist der Grenzwert der mittlere Geschwindigkeit vm (t, Δt), für Δt -> 0 , falls dieser Grenzwert existiert.

Limit (vm (t, Δt), Δt, 0 **)** $\Rightarrow \frac{9\, t^{5/4}}{8}$

Es gilt also : $v(t) = \frac{9\, t^{5/4}}{8}$



INTERPRETATION
Die momentane Geschwindigkeit (momentane Änderungsrate des Weges) zum Zeitpunkt t ist gleich der **STEIGUNG DER TANGENTE AN DER STELLE t** des Graphen von s.

Momentane Beschleunigung:
Die momentane Beschleunigung a (t) zum Zeitpunkt t ist der Grenzwert der mittleren Änderungsrate der Geschwindigkeit v (t) für Δt -> 0.

$\frac{9\, tt^{5/4}}{8}$ -> v (tt) $\Rightarrow$ Done **Limit (** (v (t+ Δt)-v(t)) / Δt , Δt, 0 **)** $\Rightarrow \frac{45\, t^{1/4}}{32}$

Beachte: Für die Definition und den Aufruf einer Funktion darf nicht die gleiche Variable verwendet werden (Error: Circular definition). Wir verdoppeln daher bei der Definition einer Funktion jeweils den letzten Buchstaben der Variablennamen.

Es gilt also : $a(t) = \frac{45\, t^{1/4}}{32}$

PROBLEM: Raketenstart

Ein Raketenstart kann mit der Funktion h : $t \mapsto t^3 / 6$ [m] beschrieben werden. Welche Geschwindigkeiten erreicht die Rakete während der Startphase? Wie steht es mit der Beschleunigung ?

CAS - Befehle

Grenzwert : **limit (** Funktionsterm, Variable, Punkt [, Richtung] **)**

Beispiel: $f(x) = \begin{cases} x & \text{falls } x \le 1 \\ x+1 & \text{sonst} \end{cases}$

when (x ≤ 1 , x, x+1**) ->** f (x)		$\Rightarrow$	Done
linksseitiger Grenzwert:	**limit (** f (x), x , 1, -1**)**	$\Rightarrow$	1
rechtsseitiger Grenzwert:	**limit (** f (x), x , 1, 1**)**	$\Rightarrow$	2
beidseitiger Grenzwert:	**limit (** f (x), x , 1**)**	$\Rightarrow$	undef

Differenzenquotient: **avgRC (** Funktionsterm, Variable [, h] **)** (**av**arage **r**ate of **c**hange)
Wird h weggelassen, so wir für h der Wert 0.001 eingesetzt.

Beispiel: $\frac{9\, t^{5/4}}{8}$ -> v (t) $\Rightarrow$ Done

avgRC (v (t), t, Δt **)** $\Rightarrow \frac{9 \cdot ((t+\Delta t)^{5/4} - t^{5/4})}{8 \cdot \Delta t}$

2.2 Grundlagen

THEORIE (Traditionelle Bearbeitung)

Grundlagen:
- Differenzenquotient - mittlere Änderungsrate
- Differentialquotient - die lokale Änderungsrate
- Ableitungen der elementaren Funktionen
- Linearität der Ableitung
- Ableitungsfunktion
- Beziehung: Ableitung - Stammfunktion
- n-te Ableitung

Grundfertigkeiten:
- Ableitungen einfacher Funktionen mathematisch herleiten.
- Ableitungsfunktion skizzieren.
- Linearität bei der Ableitung von Funktionen einsetzen.
- Stammfunktionen einfacher Funktionen bestimmen.

Bemerkungen:
- Das CAS sollte man jeweils erst einsetzen, wenn die SchülerInnen den Grundgedanken erfasst, und beim Lösen einfacher Probleme eine gewisse Grundfertigkeit erreicht haben.
- Auf die Ableitungsregeln (Produkt- , Quotienten- und Kettenregel) könnte man verzichten. Man darf aber dann keine grosse manuelle Fertigkeit bei der Bestimmung von Stammfunktionen erwarten.

Bei der Erarbeitung der Grundlagen kann der TI-92 zur Veranschaulichung und zur Vertiefung der Theorie eingesetzt werden.

2.2.1 CAS - Ableitungen

CAS -Herleitung: Ableitungen elementarer Funktionen

ALLGEMEIN :

Gegeben : Funktion $f : x \mapsto y$

Funktionsterm -> f (xx)	⇒ Done	**Definition der Funktion f**
(f (x + δx)-f (x)) / δx -> q (x , δx)	⇒ Done	**Differenzenquotient - mittlere Änderungsrate**
Limit (q (x , δx), δx , 0 **)**		**Ableitung: Differentialquotient - lokale Änderungsrate**

Bemerkungen
- Für die Definition und den Aufruf einer Funktion darf nicht die gleiche Variable verwendet werden , wenn der Aufruf nicht identisch mit der Definition ist (Error: Circular definition). Wir verdoppeln daher bei der Definition einer Funktion jeweils den letzten Buchstaben der Variablennamen.
- δ erhält man mit [G] [↑] [d][2]. Vereinfachung : dx verwenden

BEISPIELE:

- $f(x) = x^n$
  xx^n -> f (xx) ⇒ Done
  (f (x + δx) - f (x)) / δx -> q (x , δx)
  Limit (q (x , δx), δx , 0 **)** ⇒ $\frac{n\,x^n}{x}$
  $\mathbf{f'(x) = (x^n)' = n\,x^{n-1}}$
- $f(x) = \sin x$
  ***sin*(** xx **)** -> f (xx) ⇒ Done
  (f (x +dx)-f (x)) / dx -> q (x , dx)
  Limit (q (x , dx), dx , 0 **)** ⇒ cos (x)
  $\mathbf{f'(x) = (\sin x)' = \cos x}$
- $f(x) = x^{p/r}$
- $f(x) = a\hat{}(x)$
- $f(x) = \cos(x)$
- $f(x) = \ln(x)$

Aufgaben:
CAS Herleitungen für die Ableitungen von :
tan (x) , arcsin (x), arccos (x), arctan (x), log (x)

[2] Vorsicht: Δx ist eine Systemvariable. Die Verwendung der Variablen kann zu Problemen führen.

Demonstrationsbeispiel für das Programmieren einer einfachen Funktion:

Funktion erzeugen:

- Taste [APPS]   7: Program Editor   3: New...   Type: Program -> Function
  Variable: ableiten

- Das Programm eintippen

```
ableiten ( funk, xv )
func
  local  dax, xx
  return limit ( ((funk |xv=xx+dx)- ( funk |xv =xx))/dx, dx, 0 )
endfunc
```

[HOME] Aufruf des Programms
PARAMETER:
funk : Funktionsterm f (xv)
xv: Funktionsvariable

Aufruf:
ableiten (x*sin(x),x) ⇒ xx· cos(xx)+ sin(xx)

CAS - Ableitungsbefehl *d* (f (x) , x)

Für die Ableitung muss man die Tasten 2nd [*d*] verwenden.

***d*(sin** (x) , x **)** ⇒ cos (x)

sin (b * t) * *e*^(- a * t) ->[3] w (t, a , b) ⇒ Done
***d* (** w(t, a, b) , t **)** ⇒ b cos (b t) e^{-at} - a sin (b t) e^{-at}

ANWENDUNG: Bestimme den Schnittwinkel der Graphen von f: $x \mapsto e^{-x}$ und $g: x \mapsto \frac{x}{x^2+1}$

e^(- xx) -> f (xx) ⇒ Done
xx / (xx^2 + 1) -> g (xx) ⇒ Done

zeros (f (x) - g (x) , x **) ->** x0
x0[1] -> xo

d(f (x) ,x ***)*** | x = xo -> mf;
d(g (x) ,x ***)*** | x = xo-> mg

TAN^{-1} **(abs (** (mf-mg) / (1 -mf*mg))) ⇒ 31.4895
bzw. Im Bogenmass ⇒ 0.549595

mf * (x - xo) + f (xo) -> tf (x) ⇒ Done
mg * (x - xo) + g (xo) -> tg (x) ⇒ Done



[WINDOW] xmin= - 4 xmax=4
ymin= - 1 ymax= 1
[HOME]
graph { f (x) , g (x) , tf (x) , tg (x) }

Bemerkung: Die Tangenten können auch mit [F5] Math A : Tangent im Grafikbildschirm gezeichnet werden

[3] Es lohnt sich alle Parameter anzugeben: **Σ (** v^i, i, 0, n) -> f (n) und f (∞) | v= 3 / 4 ⇒ undef
Σ (v^i, i, 0, n) -> g (v,n) und g (v , ∞) | v= 3 / 4 ⇒ 4

CAS - Ableitungsfunktion

Der Graph der Funktion f: x $\mapsto$ $x^3 + x^2 - 6x$ und der Graph der Ableitungsfunktion f ' sollen im gleichen Koordinatensystem dargestellt werden.



1. Variante

ClrGraph

x^3 + x^2 -6*x -> f (x) $\Rightarrow$ Done

d **(** f (x) , x**)** -> df (x) $\Rightarrow$ Done

[WINDOW] xmin= - 4 xmax=3

ymin= - 8 ymax=10

[HOME] **graph** { f (x) , df (x) }

Problem: Einige Operationen in Menü [F5] [MATH] ergeben fälschlicherweise „NO SOLUTION“



2. Variante

ClrGraph

[Y =] y1(x) = x^3 + x^2 -6*x [F6] 4: Thick

y2(x) = ***d*** **(** y1(x) , x **)** [F6] 2: Dot

[GRAPH]

Aufgaben

Stelle die Graphen der Funktion f und ihre Ableitungsfunktion f ' mit dem TI-92 dar.

a) $f(x) = x^2 - 2x - 4$ b) $f(x) = ½x^3 - 5x + 2$ c) $f(x) = \sin x + 1$

Definiere mit dem CAS die Funktion f, bestimme deren Ableitungfunktion f ' und stelle die beiden Graphen dar.

a) $f(x) = \begin{cases} -x & \text{falls } x < 0 \\ x^2 & \text{falls } 0 \le x \le 2 \\ 4x - 4 & \text{falls } x > 2 \end{cases}$ b) $f(x) = \begin{cases} -x & \text{falls } x < 0 \\ \sin(x) & \text{falls } 0 \le x \le \pi \\ (x - \frac{3}{2}\pi)^2 - \frac{\pi^2}{4} & \text{falls } x > \pi \end{cases}$

Kontrolliere mit dem CAS die Lösungen der folgenden Differentialgleichungen :

a) $y' = -a\,y$ $y = b\,e^{-a x}$ b) $(y')' = y'' = -y$ $y = A \sin x + B \cos x$

2.2.2 CAS - Höhere Ableitungen *d*(f (x) , x , n)

Bemerkung: Die n-te Ableitung einer Funktion kann nicht allgemein aufgerufen werden.

Untersuche die Ableitungen von f : $x \mapsto \dfrac{x}{1-x}$

$\dfrac{xx}{1-xx}$ -> f (xx) $\Rightarrow$ Done

d(f (x) , x , 1) $\Rightarrow$ $\dfrac{1}{(x-1)^2}$ ***d***(f (x) , x , 2) $\Rightarrow$ $\dfrac{-2}{(x-1)^3}$

d(f (x) , x , 3) $\Rightarrow$ $\dfrac{6}{(x-1)^4}$ ***d***(f (x) , x , 4) $\Rightarrow$ $\dfrac{-24}{(x-1)^5}$

Vermutung : $f^{(n)}(x) = \dfrac{d^n}{dx^n} f(x) = \dfrac{(-1)^{n+1}\, n!}{(x-1)^{n+1}}$

Beweis mit vollständiger Induktion:

(-1)^(nn+1) * nn ! / (xx-1)^(nn+1)-> af (xx, nn) $\Rightarrow$ Done

Verankerung:

af (x , 0) $\Rightarrow$ $\frac{-1}{x-1}$ **falsch !**

af (x , 1) $\Rightarrow$ $\frac{1}{(x-1)^2}$ **richtig !**

Vererbung:

af (x , n + 1) - **comdenom(** ***d*** **(**af (x , n) , x**))** $\Rightarrow$ 0 **richtig !**

dh. $f^{(n+1)}(x) = (f^{(n)}(x))' \quad \forall \; n \geq 1$

Aufgaben

Bestimme die n-te Ableitung von f und beweise die Vermutung.

a) $f: x \mapsto x\,e^{x}$ b) $f: x \mapsto \sqrt{2x+1}$ c) $f: x \mapsto \sin x \; \cos x$ e) $f: x \mapsto \frac{x-1}{x+1}$

Erweiterungen:

- **CAS - Einseitige Ableitungen**

  abs (sin (xx **))** -> f (xx) mittlere Änderungsrate: **avgRC (** f (x), x , h **)** $\Rightarrow$ $\frac{|\sin(x+h)| - |\sin(x)|}{h}$

  linksseitige Ableitung: **limit (avgRC(**f (x), x, h), h, 0, -1 **))** | x = 0 $\Rightarrow$ - 1

  rechtsseitige Ableitung: **limit (avgRC(**f (x), x, h), h, 0, 1 **))** | x = 0 $\Rightarrow$ 1

- **CAS - numerische Ableitung**

  nDeriv (f (x) , x , h **)** $\Rightarrow$ $\frac{-(|\sin(x-h)| - |\sin(x+h)|)}{2h}$

2.2.3 CAS - UNTERSUCHUNG zur *Produktregel*

Leite mit dem CAS ab :
$\sin x \; e^{x}$ $\Rightarrow$ ____________________________

Folgerung: Welche naheliegende Regel gilt sicher **nicht** ?

____

____

Bestimme mit dem CAS die Ableitungen der folgenden Funktionen.

Funktion:	Ableitung:
$h(x) = x^3 \sin x$ $h(x) = x^2 e^x$ $h(x) = \cos x \; \ln x$	
$h(x) = \sqrt[3]{x^2}\, e^x$	

Finde und formuliere eine Regel. Zerlege dazu h in 2 Funktionen f und g .

VERSUCH: ____

____

Produktregel:

Überprüfe die Vermutung an den folgenden Beispielen:

Funktion:	Vermutung (Regel)	Ableitung mit CAS
$h(x) = x^3 \ln x$ $h(x) = x^2 \cos x$ $h(x) = \cos x \; \sin x$		
$h(x) = \sqrt[5]{x^3} \; x^4$		

Mathematischer Beweis:

____

____

Bemerkungen:
limit (...) funktioniert nicht für $f(x)\,g(x)$
d **(** f(x)*g(x)), x **)** liefert direkt die Produktregel, aber das ist natürlich kein Beweis

Die Quotienten- und Ketttenregel können analog eingeführt werden.

2.2.4 **CAS** - UNTERSUCHUNG zur Differenzierbarkeit

Es gibt Funktionen, die sind an einzelnen Stellen definiert, aber dort nicht differenzierbar. Die Sekantenbüschel bei differenzierbaren Stellen sind grundsätzlich verschieden von den Sekantenbüschel bei nicht differenzierbaren Stellen.

Programm erzeugen:

- Taste [APPS]   7: Program Editor   3: New...   Type: Function -> Program
  Variable: Diffbar

- Das Programm eintippen

```
Diffbar ( funk, x, x0)
Prgm

Local deltas, i, dx, m, y0

{2,1.75,1.5,1.25, 1,.75,.5, .25, 01}->deltas

x0-3.5->xmin : x0+3.5 -> xmax
-1.5+ funk |x=x0 -> ymin
1.5+ funk |x=x0  -> ymax

funk |x=x0->y0
ClrGraph : ClrDraw

DrawFunc  funk
Circle  x0, y0, .1

For i, 1, 8
   deltas[i]->dx
   (( funk | x=x0+dx ) - y0 ) / dx -> m
   DrawSlp x0, y0, m

   -dx->dx
   (( funk | x=x0+dx ) - y0 ) / dx -> m
   DrawSlp x0, y0, m
EndFor

EndPrgm
```

Aufruf: (im Home-Screen)

[HOME]

Diffbar (ABS(sin(x)) , x , 0)   [ENTER]

PARAMETER:
funk : Funktionsterm f (x)
x : Funktionsvariable
x0 : Stelle



Aufgaben
Untersuche mit dem Programm die Differenzierbarkeit von
a) $f(x) = | 1 / 4 x^2 - 4 |$
b) $f(x) = \sin x + | \sin x |$
Was bedeutet die Differenzierbarkeit an der Stelle x_0 anschaulich ?

Hinweis: In einem Programm kann die Strichart nicht ausgewählt werden.

Quelle: Scheuermann Hellmut und Weller Hubert, Computersoftware als methodische Hilfe bei der Begriffsentwicklung im Mathematikunterricht, erschienen in: Hirscher Horst (Herausgeber), Wieviel Termumformung braucht der Mensch?, Verlag Franzbecker, Hildesheim, 1993

2.3 Kurven

2.3.1 Kurvendiskussion mit dem CAS

Untersuchung am Beispiel $f(x) = 1/8 \cdot x^3 - ½ \cdot x^2 - 2½x + 6$

Definition: 1 / 8 * xx^3 - 1/ 2 * xx^2 - 5 / 2 * xx + 6 -> f (xx) ⇒ Done

Nullstellen: **zeros (** f (x), x **)** ⇒ { - 4 2 6 }

Extrema: Illustration des Satzes: An den Extremalstellen des Graphen von f gilt f '(x) = 0.

Wie verlaufen die Tangenten in den Extremalpunkten?
graph f (x)
Suche Minimum und Maximum und zeichne dort die Tangenten ein:

[F5] / 3: Minimum
Lower Bound? xc: 3 / Upper Bound? xc: 5
⇒ Minimum (4.23927 / –4.06067)

[F5] / [A]: Tangent Tangent at?
xc: 4.23927 ⇒ y=0.000009x–4.06071



[F5] / 4: Maximum
Lower Bound? xc: –3 / Upper Bound? xc: –1
⇒ Maximum (–1.5726 / 8.20882)

[F5] / [A]: Tangent Tangent at?
xc: –1.5726 ⇒ y=0.000002x+8.20882



Vermutung: Bei den Extrema von f ist die Tangentensteigung 0, also f '(x)=0.
Überlegung: f'(x)=0 ist sicher notwendige Bedingung für eine Extremalstelle.

Anwendung des Satzes zur exakten Berechnung der Extrema:

◆ [Home]

zeros(***d(*** f (x), x), x**)** → extrema $\Rightarrow \left\{ \frac{-2(\sqrt{19}-2)}{3} \quad \frac{2(\sqrt{19}+2)}{3} \right\}$

ans(1**)** ◆[Enter] ⇒ {–1.5726 4.23927}

Berechnung der Funktionswerte an den gefundenen Stellen:

f (extrema) $\Rightarrow \left\{ \frac{2\left(19\sqrt{19}+28\right)}{27} \quad \frac{-2\left(19\sqrt{19}-28\right)}{27} \right\}$

ans(1**)** ◆[Enter] ⇒ {8.20882 –4.06067}

Wendepunkte: Klassische Herleitung des Satzes: In den Wendepunkten des Graphen von f gilt $f''(x) = 0$.

Anwendung des Satzes zur exakten Berechnung des Wendepunktes:

zeros(***d*** **(** f (x), x, 2 **)**, x**)** → wendept ⇒ {4/3}
f (wendept) ⇒ {56/27}
ans(1) ◆[Enter] ⇒ {2.07407}



Direkte Bestimmung:

Graph f (x)
[F 5] / [8]: Inflection
Lower Bound ? xc: 0 [Enter]
Upper Bound ? xc: 4 [Enter]

Wendetangente:
◆ [Home]
(***d*** (f (x) , x**)** | x=4/3)*(x-4/3)+56/27-> t (x) ⇒ Done
Graph t (x)

Programm : Kurvendiskussion

Die Kurvendiskussion kann weitgehend automatisiert werden. Mit dem folgenden einfachen Programm können rationale Funktionen diskutiert werden.

Aufruf : (im Home-Screen)

◆ [HOME]

[MODE] [F2] Split Screen .... FULL -> 3: LEFT-RIGHT

◆ [WINDOW]
xmin = -5 xmax = 5
ymin = -3 ymax = 3

kurvdisk ((x^2 - x) / (x^2 -x - 6), x)



Programm erzeugen:

- Taste [APPS] 7: Program Editor 3: New... Type: PROGRAM ->
  Variable: kurvdisk

- Das nachfolgende Programm eintippen

```
kurvdisk ( funk, xv )
Prgm

   Local  xx, zz

   ClrIO
   ClrGraph :     ClrDraw
   DrawFunc funk |xv=x
   Disp   "Weiter : Taste"
   While  getKey()=0 :  EndWhile

   ClrIO
   zeros ( funk, xv ) -> xx
   Disp   "NULLSTELLEN:"
   Disp   "----------------------"
   Disp   xx
   Disp   " "
   Disp   "Weiter : Taste"
   While getKey()=0 :  EndWhile

   ClrIO
   zeros ( d (funk, xv ), xv ) -> xx
   Disp   "EXTREMA:"
   Disp   "----------------"
   Disp   "x-Werte  "&string ( xx )
   Disp   "y-Werte  "&string ( funk |  xv =xx )
   d ( funk ,xv, 2 ) | xv=xx   -> zz
   Disp   "Min / Max  "&string ( zz )
   Disp   "+1 : Max -1: Min     sign (0) :   Sattel"
   Disp   " "
   Disp   "Weiter : Taste"
   While  getKey()=0 :  EndWhile
```

```
ClrIO
zeros ( d (funk, xv, 2 ), xv ) -> xx
Disp   "WENDEPUNKTE:"
Disp   "------------------------"
Disp   "x-Werte  "&string ( xx )
Disp   "y-Werte  "&string ( funk |  xv =xx )
Disp   " "
Disp   "Weiter : Taste"
While  getKey()=0 :  EndWhile


ClrIO
Disp   "Gebr. rat. Funktion"
Disp   "------------------------"
Disp   "ASYMPTOTEN:"
Disp   "y= .....    ( ganzer Teil von : )"
Disp   propFrac ( funk )
Disp   " "
Disp   "Weiter : Taste"
While  getKey()=0 :  EndWhile

ClrIO
Disp "Gebr. rat. Funktion"
Disp "------------------------"
Disp "POLE:"
Disp "x = "&string ( zeros(getDenom ( funk ), xv ))
Disp " "
Disp "Weiter : Taste"
While  getKey()=0 :  EndWhile


EndPrgm
```

PARAMETER:
funk : Funktionsterm
xv : Funktionsvariable

2.3.2 Interpolation mit Ableitungen



Beispiel: Strassenbau
Die beiden Strassen s_1 und s_2 sollen durch ein möglichst optimales Zwischenstück miteinander verbunden werden.
s_1: $y=0$, $x\le 0$, s_2: $y=x-3$, $x\ge 6$, A (0, 0), B (6, 3).

1. Ansatz: Verbinde A und B durch ein Geradenstück
Gerade durch A und B:

{0, 6} → xwerte ⇒ Done
{0, 3} → ywerte ⇒ Done
linreg xwerte, ywerte
regeq(x) → y(x) ⇒ Done
y(x) ⇒ .5 · x

Dieser Weg ist nur möglich, weil der TI-92 die passende Regressionsmethode zur Verfügung stellt.

Praktischer Nachteil: Knicke bei A und B. Bei hoher Geschwindigkeit besteht erhebliche Unfallgefahr!

2. Ansatz: Verbinde A und B durch den Graphen einer Polynomfunktion:
Folgende Bedingungen sind zu erfüllen:

Anschluss bei A:	$y(0) = 0$
Anschluss bei B:	$y(6) = 3$
Kein Knick bei A:	$y'(0) = 0$
Kein Knick bei B:	$y'(6) = 1$
Keine abrupte Bewegung des Lenkrades bei A:	$y''(0) = 0$
Keine abrupte Bewegung des Lenkrades bei B:	$y''(6) = 0$

Es sind 6 Parameter nötig, weshalb wir mit einem Polynom 5. Grades arbeiten:

a*x^5+b*x^4+c*x^3+d*x^2+e*x+f → y(x)	⇒ Done
y(0)=0	⇒ f=0
y(6)=3	⇒ 7776a + 1296b + 216c + 36d + 6e + f = 3
(***d***(y(x),x) |x=0)=0	⇒ e=0
(***d***(y(x),x) |x=6)=1	⇒ 6480a + 864b + 108c + 12d + e = 1
(***d***(y(x),x,2) |x=0)=0	⇒ 2d=0
(***d***(y(x),x,2) |x=6)=0	⇒ 4320a + 432b + 36c + 2d = 0

Nach manuellem Vereinfachen bleibt noch folgendes Gleichungssystem übrig:

$$\begin{aligned} 7776a + 1296b + 216c &= 3 \\ 6480a + 864b + 108c &= 1 \\ 4320a + 432b + 36c &= 0 \end{aligned}$$

Lösung mit **simult** oder **rref** liefert a = 0, b = –1/432, c = 1/36, also $y(x) = -\frac{1}{432}x^4 + \frac{1}{36}x^3$.

Der Unterricht und die zu verwendenden Lösungsverfahren sind bei diesem Thema stark rechnerabhängig. Bei einigen CAS genügt es, den Ansatz für y(x) festzulegen und die Bedingungsgleichungen zu formulieren. Den Rest erledigt der solve-Befehl.

Aufgaben:

1. Strassenbau I: Die Autobahnen g: $y = x$ und h: $y = -x$ sollen durch ein möglichst sicheres Strassenstück miteinander verbunden werden. Das Strassenstück führt von

   a) A(1, 1) nach B(–1, 1) b) A nach C (1 ,-1)

   Bestimmen Sie die Gleichung des Verbindungsstück in den Fällen a) und b).

2. Strassenbau II: Die Strassen s_1: $y = -2$, $x \le -2$ und s_2: $y=2$, $x \ge 2$ sollen durch ein s-förmiges Stück miteinander verbunden werden. Welches ist die Gleichung dieses Verbindungsstücks?

3. Die folgende Aufgabe ist etwas speziell: Weder bei a) noch bei b) treten Ableitungen auf. Dafür führt b) auf ein nichtlineares Gleichungssystem, dessen Lösung mit dem TI-92 einige Geduld erfordert.
   Ein Kabel soll über eine 10m breite Strasse gehängt werden. Die Befestigungspunkte befinden sich auf beiden Seiten in 8m Höhe, und in der Strassenmitte ist das Kabel noch 6m über der Strasse.

   a) Wir nehmen an, der Verlauf des Kabels werde durch ein Polynom beschrieben. Welches ist seine Gleichung? Wie lange muss das Kabel sein (das können Sie im GRAPH-Bildschirm mit F5 / B: Arc berechnen lassen)?

   b) Die Physik lehrt, dass der Verlauf des Kabels besser durch eine Funktion der Bauart $a \cdot \cosh(x / a) + b$ beschrieben wird. Welches ist seine Gleichung? Wie lange muss das Kabel sein (das können Sie im GRAPH-Bildschirm mit F5 / B: Arc berechnen lassen)?

   c) Stellen Sie die beiden resultierenden Graphen dar.

   Hinweis: Wählen Sie den Nullpunkt des Koordinatensystems genau in der Strassenmitte.

2.3.3 Ellipsengleichung



Definition :Ellipse = { P (x, y) | $\overline{F_1P} + \overline{F_2P} = 2\cdot a$ }

CAS-Herleitung:

Quadrieren:
expand ((√ (y^2+(c-x)^2) + √ (y^2+(c+x)^2) = 2*a)^2**)**

$$\Rightarrow 2\cdot\sqrt{x^2-2\cdot c\cdot x+y^2+c^2}\cdot\sqrt{x^2+2\cdot c\cdot x+y^2+c^2}+2\cdot x^2+2\cdot y^2+2\cdot c^2=4\cdot a^2$$

Wurzeln separieren:
ans (1) -2*x^2 -2*y^2-2*c^2

$$\Rightarrow 2\cdot\sqrt{x^2-2\cdot c\cdot x+y^2+c^2}\cdot\sqrt{x^2+2\cdot c\cdot x+y^2+c^2}=-2\cdot x^2-2\cdot y^2+4\cdot a^2-2\cdot c^2$$

Quadrieren:
expand (ans (1) ^2 **)** -> gleich

$$\Rightarrow 4\cdot x^4+8\cdot x^2\cdot y^2-8\cdot c^2\cdot x^2+4\cdot y^4+8\cdot c^2\cdot y^2+4\cdot c^4=4\cdot x^4+8\cdot x^2\cdot y^2-16\cdot a^2\cdot x^2+8\cdot c^2\cdot x^2+4\cdot y^4-16\cdot a^2\cdot y^2+8\cdot c^2\cdot y^2+16\cdot a^4-16\cdot a^2\cdot c^2+4\cdot c^4$$

Ordnen:
0 = **right (** gleich **) - left (** gleich **)**

$$\Rightarrow \quad 0=(-16\cdot a^2+16\cdot c^2)\cdot x^2-16\cdot a^2\cdot y^2+16\cdot a^4-16\cdot a^2\cdot b^2$$

Ersetzen:
ans (1) **|** c^2=a^2 - b^2 $\Rightarrow$ $0=-16\cdot b^2\cdot x^2-16\cdot a^2\cdot y^2+16\cdot a^2\cdot b^2$

Vereinfachung:

propFrac (ans (1) / (16*a^2*b^2)*(-1) +1 **)** $\Rightarrow$ $1=\frac{x^2}{a^2}+\frac{y^2}{b^2}$ oder nur

ans (1) / (-16) + a^2*b^2 $\Rightarrow$ $a^2\cdot b^2=b^2\cdot x^2+a^2\cdot y^2$

Implizites Differenzieren:

***d* (** x^2 / a^2 + y(x)^2 / b^2 = 1, x **)** $\Rightarrow$ $\frac{2\cdot y(x)\cdot\frac{d}{dx}(y(x))}{b^2}+\frac{2\cdot x}{a^2}=0$

Warning: Differentiating an equation may produce a false equation

Tangentengleichung:

(**ans** (1) **-** 2*x / a^2) * b^2 / (2*y(x)) $\Rightarrow$ $\frac{d}{dx}(y(x))=\frac{-b^2\cdot x}{a^2\cdot y(x)}$

right (ans(1)**)** -> m(x) $\Rightarrow$ Done

y(x) - y(x0) - m(x0)*(x-x0)=0 $\Rightarrow$ $y(x)+\frac{b^2\cdot x0\cdot x-a^2\cdot(y(x0))^2-b^2\cdot x0^2}{a^2\cdot y(x0)}=0$

ans (1) *a^2*y(x0) $\Rightarrow$ $a^2\cdot y(x0)\cdot y(x)+b^2\cdot x0\cdot x-a^2\cdot(y(x0))^2-b^2\cdot x0^2=0$
Manuelle Vereinfachung: $\Rightarrow$ $a^2\cdot y(x0)\cdot y(x)+b^2\cdot x0\cdot x=a^2\cdot b^2$

Erweiterung:
Krümmung und Krümmungsradius

2.4 Extremalwertprobleme

2.4.1 Zwei geometrische Extremalwertprobleme

1. PROBLEM
Aus gleichlangen Stangen (Länge a) werden die Stützen für ein Zelt hergestellt. Wie baut man damit ein optimales Zelt ? Hast du ein Vermutung für die Lösung ?

Modellierung:
Das Zelt ist etwa optimal, wenn die Querschnittfläche des Zeltes (Dreiecksprisma) maximal ist.



$f(x, h) = x\,h$ mit $h = \sqrt{a^2 - x^2}$
Bedingungen: $0 \le x \le a$

CAS - Lösungen:

√ (aa^2 - xx^2) -> h ⇒ Done (Nebenbedingung)
xx * h -> f (xx , aa) ⇒ Done (Extremalbedingung)

Ableiten und Nullstllen bestimmen:
zeros (***d*** **(** f (x , a), x **)** , x **)** ->xm ⇒ $\left\{\frac{|a|}{\sqrt{2}} \quad \frac{-|a|}{\sqrt{2}}\right\}$

Kontrolle:
d **(** f (x , a), x **)** | x = xm[1]  [Enter] ⇒ 0.
d **(** f (x , a), x ,2**)** | x = xm[1]  [Enter] ⇒ - 4 rel. Maximum

Man muss die Grundlinie $a\sqrt{2}$ wählen, damit die Querschnittfläche maximal ist.
(Die 2.Lösung ist geometrisch sinnlos)

Direkte Bestimmung des Maximum:

fmax (f (x , a) , x **)** | a > 0 ⇒ $x = \frac{a}{\sqrt{2}}$

Lösung für einen konkreten Wert a = 2:

 [WINDOW] xmin = 0 xmax = 3
ymin = 0 ymax = 3 xres = 1

 [HOME]
graph f (x, 2)

[F 5] 4: Maximum
Lower Bound 1 Upper Bound 2

2. Variante : **fmax (** f (x , a) , x **)** | a =2



Aufgaben
- Aus 4 gleich langen Stangen werden die Stützen für ein Zelt hergestellt, das die Form einer Pyramide hat. Wie baut man damit ein Zelt mit maximalem Volumen?

HINWEISE:
- Bei komplizierten Funktionen muss man manchmal zuerst die Ableitung vereinfachen, bevor man die Befehle **zeros** oder **solve** anwenden kann.
- Der Befehle **fmax** bzw. **fmin** funktionieren bei Funktionen mit Parametern oder bei komplizierten Funktionen nicht immer.
- Die Lösungen müssen **immer** überprüft werden.

2. PROBLEM

Einer Kugel mit dem Radius r ist ein Doppelkegelstumpf so einzuschreiben , dass die Oberfläche des Doppelkegelstumpfs maximal ist.



Tabellenbuch : $O = 2\,\pi\, s\,(\,r + x\,)\; + 2\,\pi\, x^2$

$h = \sqrt{r^2 - x^2}$ $\quad s = \sqrt{h^2 + (r - x)^2}$

Bedingungen: $0 \le x \le r$

CAS -Lösungen :

√ (rr^2 - xx^2 **)** **->** h	(Nebenbedingung)
√ (h^2 + (rr - xx)^2 **)** **->** s	(Nebenbedingung)
2*π*(xx^2 + s*(rr +xx)) -> o(xx , rr)	(Extremalbedingung)

Allgemeine Lösung:

fMax (o (x , r) , x **)** | x > 0 **and** x < r $\Rightarrow$ $4 \cdot x \cdot \sqrt{-(x - r) \cdot r} - \sqrt{2} \cdot (3 \cdot x - r) \cdot r = 0$ or $x = \infty$ or $x = \infty$ or $r = 0$

zeros(comdenom(*d* (o(x, r),x)**), x)** -> x0 $\Rightarrow$ { } **Fehler !!**

Lösung für r = 1:

fMax (o (x , 1) , x **)** $\Rightarrow$ x = - ∞ **!!**

o (x ,1) $\Rightarrow$ $2\pi\,((x+1)\sqrt{-2(x-1)} + x^2)$

zeros(comdenom(*d* (o(x, 1),x)**), x)** -> x0 $\Rightarrow$ { -1. . 695194 } TI-92 liefert nur *numerische* Lösungen

Lösung für r =1 mit Bereichsbegrenzung:

fMax (o (x , 1) , x **)** | x > 0 $\Rightarrow$ x =. 695194 TI-92 liefert nur eine *numerische* Lösung

Untersuchung:

comdenom (*d* (o(x , r) ,x **))** $\Rightarrow$ $\dfrac{4\pi\sqrt{r\,(r - x)}\; x + \pi\sqrt{2}\; r^2 - 3\,\pi\sqrt{2}\; r\; x}{\sqrt{r\,(r - x)}}$

comdenom (*d* (o(x , 1) ,x **))** -> o1 (x) $\Rightarrow$ Done

graph o1 (x)



Es gibt mindestens 2 reelle Lösungen :
x ≈ - 1 und 0 < x < 1

Lösungsansätze:

1. Ähnlichkeit : r = 1

zeros (o1(x) , x **)** | x > 0-> x0 $\Rightarrow$ { . 695194 }

o1 (x) | x = x0[1] $\Rightarrow$ 0.

***d* (** o(x , 1) ,x , 2 **)** | x = x0[1] $\Rightarrow$ -25.9062 rel. Max

o (x , 1) | x = x0[1] $\Rightarrow$ 11.3529

Alternative:

fMax (o (x ,1), x) | x > 0->x0 $\Rightarrow$ { . 695194 }

x0[1] * r -> rx $\Rightarrow$ Done

o (rx , r) $\Rightarrow$ $11.3529\, r^2$

2. Manuelle Umformungen

(4 √ (r*(r - x))* x = 3*r*x √ (2) - √ (2) * r ^2)^2 -> t

Warnig: Operation might introduce false solutions

zeros (left(t **) - right (** t **)** , x**)** | r > 0 **and** x > 0 -> x0

$\Rightarrow \left\{ \dfrac{-\,(\sqrt{17} - 7)\; r}{16} \quad \dfrac{(\sqrt{17} + 7)\; r}{16} \right\}$

***d* (** o(x , r) ,x **)** | x = x0[2] **and** r > 0 $\Rightarrow$ - 8. E -13 r

***d* (** o(x , r) ,x **)** | x = x0[1] **and** r > 0 $\Rightarrow$ 4.51902 r

x0[1] ist keine Lösung !

***d* (** o(x , r) ,x , 2) | x= x0[2] **and** r > 0 $\Rightarrow$ -25.9062 rel. Max

o (x0[2] , r) $\Rightarrow$ $11.3529\, r^2$

Feststellungen: Die Aufgabe kann **nur** mit geometrischen oder algebraischen Kenntnissen gelöst werden. Wurzelgleichungen bereiten dem TI-92 grosse Schwierigkeiten.

2.4.2 Zwei Extremalwertprobleme aus der Physik

1. PROBLEM
Mit welcher Geschwindigkeit muss man sich abstossen, um mit min. Energie die Höhe y0 im Abstand x0 zu erreichen ? (Nach [11])
Beispiel: x0 = 4 und y0 = 3



Tabellenbuch (Physik)

Schiefer Wurf: $y = x \cdot \tan \alpha_0 - \dfrac{g}{2\, v_0^2 \cos^2 \alpha_0} \cdot x^2$

CAS -Lösung:

Wir verwenden zur Vereinfachung : $t = \alpha_0$

solve (y0=x0***tan**(t) - g*x0^2 / (2*v^2* **cos** (t)^2), v **)** $\Rightarrow$ $v = \dfrac{\sqrt{2\cdot g}\cdot x0}{2\cdot\sqrt{\cos(t)}\cdot\sqrt{-(y0\cdot\cos(t)-x0\cdot\sin(t))}}$ **or** ...

√ (2*g) *x0 / (2*√ (**cos**(t)) *√ (-(y0***cos**(t)-x0***sin**(t))) -> v(t) $\Rightarrow$ Done

Lösungsversuche:

fMin (v (t) ,t **)** $\Rightarrow$ $\cos(t) \cdot (x0 \cdot \cos(t) + y0 \cdot \sin(t)) = \dfrac{x0}{2}$ **or** ...

zeros (*d* (v (t) ,t **)** , t) $\Rightarrow$ { }
Warning: Solve might specify more zeros

solve(*d* (v (t) ,t **) =** 0 , t) $\Rightarrow$ $\cos(t) \cdot (x0 \cdot \cos(t) + y0 \cdot \sin(t)) = \dfrac{x0}{2}$ **or** ...

Ersetzung der Variablen:

ans (1**) | cos (**t**)** =z **and sin (**t**)** =√ (1- z^2) $\Rightarrow$ $2 \cdot y0 \cdot z \cdot \sqrt{-(z^2-1)} + x0 \cdot (2 \cdot z^2 - 1) = 0$ **or** ...

solve (ans (1**) , z)** $\Rightarrow$ ***keine Auflösung !!*** *Wurzelproblem des TI-92*

Manuelle Hilfe:
(2· y0· z· √ (- (z^2 - 1)) = -x0 · (2· z² -1)) | ^2 $\Rightarrow$ $-4 \cdot y0^2 \cdot z^2 \cdot (z^2-1) = x0^2 \cdot (2 \cdot z^2 - 1)^2$
Warning: Operation might introduce false solutions

solve (ans (1**) , z)** $\Rightarrow$ $z = \dfrac{\sqrt{2\cdot(\sqrt{x0^2+y0^2}+y0)}}{2\cdot(x0^2+y0^2)^{1/4}}$ **or**

$z = \dfrac{-\sqrt{2\cdot(\sqrt{x0^2+y0^2}+y0)}}{2\cdot(x0^2+y0^2)^{1/4}}$ ***sicher keine Lösung!***

Kontrolle:
(2· y0· z· √ (- (z^2 - 1)) + x0 · (2· z² -1)) | z =√ 2* (√ (x0^2+y0^2) +y0)) / (2*(x0^2+y0^2)^(1/4))

$$\Rightarrow \quad \frac{\sqrt{\sqrt{x0^2+y0^2}+y0}\cdot\sqrt{\sqrt{x0^2+y0^2}-y0}\cdot y0}{\sqrt{x0^2+y0^2}}+\frac{x0\cdot y0}{\sqrt{x0^2+y0^2}} \neq 0$$

z ist keine Lösung !

Feststellung : Der solve-Befehl unterschlägt die Lösung $z = \dfrac{\sqrt{2\cdot(\sqrt{x0^2+y0^2}-y0)}}{2\cdot(x0^2+y0^2)^{1/4}}$. Das CAS betrachtet sie als komplexe Lösung.

Kontrolle: (2· y0· z· √ (- (z^2 - 1)) + x0 · (2· z² -1)) | z =√ 2* (√ (x0^2+y0^2) - y0)) / (2*(x0^2+y0^2)^(1/4))

$$\Rightarrow \quad \frac{\sqrt{\sqrt{x0^2+y0^2}+y0}\cdot\sqrt{\sqrt{x0^2+y0^2}-y0}\cdot y0}{\sqrt{x0^2+y0^2}}-\frac{x0\cdot y0}{\sqrt{x0^2+y0^2}}$$

z ist eine Lösung !

Hinweis: Dass der Term 0 ist, muss man mit manuellen Umformungen zeigen.

Minimum:

√ (2* (√ (x0^2+y0^2) - y0)) / (2*(x0^2+y0^2)^(1/4)) ->z	⇒	Done
d **(** v (t) ,t ,2**)**		
ans (1) | **cos** (t) =z **and** **sin** (t) =√ (1- z^2)	⇒	*sehr komplizierter Term ! Manuelle Vereinfachungen und Ersetzungen sind notwendig.*

Beispiel:

ans (1) | x0 = 4 **and** y0 = 3 ⇒ $10 \cdot \sqrt{2 \cdot g}$ **relatives Minimum**

Minimale Geschwindigkeit:

v (t) ⇒ $\dfrac{\sqrt{2 \cdot g} \cdot x0}{2 \cdot \sqrt{\cos(t)} \cdot \sqrt{-(y0 \cdot \cos(t) - x0 \cdot \sin(t))}}$

ans (1) | **cos** (t) =z **and** **sin** (t) =√ (1- z^2)

⇒ $$\frac{\sqrt{g} \cdot x0 \cdot (x0^2 + y0^2)^{1/4}}{(\sqrt{x0^2+y0^2} - y0)^{1/4} \cdot \sqrt{-(\sqrt{\sqrt{x0^2+y0^2} - y0} \cdot y0 - x0 \cdot \sqrt{\sqrt{x0^2+y0^2} + y0})}}$$

Manuelle Vereinfachung: $v_{min} = \dfrac{\sqrt{g} \cdot x0}{\sqrt{\sqrt{x0^2+y0^2} - y0}}$ *Wurzelproblem des TI-92*

Beispiel: **ans** (1) | x0 = 4 **and** y0 = 3 ⇒ $2\sqrt{2 \cdot g}$

Problem : Warum liegt der Scheitel der Sprungparabel nicht im Punkt (x0 , y0) ??

Es ist : $y_{max} = \dfrac{v^2 \cdot \sin^2 t}{2 \cdot g}$ mit

$\sin t = \sqrt{1 - z^2} = \sqrt{\dfrac{y0}{2 \cdot \sqrt{x0^2 + y0^2}} + 1/2}$ und $v = v_{min} = \dfrac{\sqrt{g} \cdot x0}{\sqrt{\sqrt{x0^2+y0^2} - y0}}$

Berechnung von y_{max} :

√ (g) * x0 / √ (√ (x0^2 + y0^2) - y0) -> vmin ⇒ Done

vmin^2*(1-z^2) / (2*g) ⇒ $\dfrac{x0^2 \cdot (\sqrt{x0^2+y0^2} + y0)}{4 \cdot \sqrt{x0^2+y0^2} \cdot (\sqrt{x0^2+y0^2} - y0)} \neq y0$

Beispiel: $y_{max} = \dfrac{16}{5} \neq y0 = 3$ ***unerwartet, erstaunlich !!!***

Optimale Lösung:	Scheitel in P (4 , 3) :
$v_{min} = 2\sqrt{2 \cdot g} \approx 2.828 \sqrt{g}$	$v_0 = \sqrt{\dfrac{26}{3} g} \approx 2.944 \sqrt{g}$
$t_{min} \approx 63.43°$	$t_0 \approx 56.31°$

Erklärung:

Wenn der Scheitel nicht im Punkt (x0 , y0) sein muss, kann der Winkel etwas grösser sein. Dadurch wird die Absprunggeschwindigkeit verkleinert.

Kommentar:

- Die allgemeine Aufgabe ist mit dem CAS **fast** lösbar.
- **fMin** ist auch beim Zahlenbeispiel keine Hilfe.
- Die bekannte Wurzelschwäche des TI-92 muss mit manuellen Umformungen überbrückt werden.
- Unschön ist, dass die richtige Lösung, die für das Beispiel gefunden wird, im allgemeinen Fall nicht angegeben wird.

2. PROBLEM:

Eine Autobahn, die mit der Geschwindigkeit v_A befahren werden kann, verläuft vom Punkt (0, 0) zum Punkt (c, 0). Eine Landstrasse, wo die Geschwindigkeit v_L gestattet ist, verläuft durch den Punkt (0, b). Die beiden Strassen sollen in jenem Punkt (x, 0) miteinander verbunden werden, dass die Fahrzeit von (0, b) nach (c, 0) minimal wird. (Nach [11])



CAS - Lösung:

Funktion Fahrzeit definieren: **√ (**b^2+x^2**)** / vl + (c-x)/va -> t(x) $\Rightarrow$ Done

Nullstellen der Ableitung : **zeros(** ***d*(** t(x), x), x**)** $\Rightarrow$ { }
Warning: Solve might specify more zeros

2. Anlauf mit solve: **solve(** ***d*(** t(x), x) = 0, x**)** $\Rightarrow$ $vl\cdot\sqrt{x^2+b^2}-va\cdot x=0$

Hilfe für den Rechner: +va*x $\Rightarrow$ $vl\cdot\sqrt{x^2+b^2}=va\cdot x$

^2 $\Rightarrow$ $vl^2\cdot(x^2+b^2)=va^2\cdot x^2$
Warning: Operation might introduce false solutions

3. Anlauf mit solve: **solve(ans(** 1 **)**, x) $\Rightarrow$ $x=-|b\cdot vl|\cdot\sqrt{\frac{-1}{vl^2-va^2}}$ **and** $\frac{b^2\cdot vl^2}{vl^2-va^2}\le 0$

or $x=|b\cdot vl|\cdot\sqrt{\frac{-1}{vl^2-va^2}}$ **and** $\frac{b^2\cdot vl^2}{vl^2-va^2}\le 0$

Diskussion:

1. vl ≥ va *keine reelle Lösung!* *Lösung:* x = c

2. $|b\cdot vl|\cdot\sqrt{\frac{-1}{vl^2-va^2}}\ge c$ $\Rightarrow$ $\frac{\sqrt{va^2-vl^2}}{vl}\le\frac{b}{c}$ *Lösung: Man wird* x = c wählen.

Beispiel: va = 120 km / h vl = 80 km / h $\Rightarrow$ $\frac{\sqrt{va^2-vl^2}}{vl}=\frac{\sqrt{5}}{2}\approx 1.12$

b = 12 km c = 10 km $\Rightarrow$ $\frac{6}{5}=1.2$

x ≈ 10.7 km *Man wird* als Lösung x = c =10 km wählen.

2.4.3 Extremalwertproblem aus der Wirtschaft

PROBLEM

Für die Lancierung eines neuen Haushaltgeräts wurden sorgfältige Untersuchungen von einem Marktforschungsinstitut und der Kalkulationsabteilung der Produktionsfirma durchgeführt. Der Verkaufspreis pro Gerät ist x Fr.

Die Marktanalyse ergab die Nachefragefunktion: $N(x) = a \cdot e^{-b\,x^2}$ $\quad a, b > 0$

Die Kalkulationabteilung berechnete die Produktionskostenfunktion: $K(x) = c + \dfrac{d}{N(x)}$ $\quad c, d > 0$

Bei welchem Preis x ist der Gewinn am grössten und wie gross ist dieser Gewinn?

CAS - Lösung:

a ***e**^(- b*x^2) -> n (x) $\Rightarrow$ Done
c + d / n (x) -> k (x) $\Rightarrow$ Done
n(x) * (x - k (x)) -> g (x) $\Rightarrow$ Done
g (x) $\Rightarrow - \left(d \cdot e^{b\cdot x^2} - a \cdot (x - c) \right) \cdot e^{-b\cdot x^2}$

Klassischer Weg:

zeros (*d* (g (x) , x **)** , x **)** ->xe $\Rightarrow \left\{ \dfrac{\sqrt{b\cdot c^2+2}+\sqrt{b}\cdot c}{2\cdot\sqrt{b}} \quad \dfrac{-(\sqrt{b\cdot c^2+2}-\sqrt{b}\cdot c)}{2\cdot\sqrt{b}} \right\}$

xe[2] < 0 ist keine Lösung.

***d* (** g (x) , x , 2**)** | x = xe [1] $\Rightarrow -2 \cdot a \cdot \sqrt{b} \cdot \sqrt{b\cdot c^2+2} \cdot e^{\frac{-\sqrt{b}\sqrt{b\cdot c^2+2}\cdot c}{2} - \frac{b\cdot c^2}{2} - 1/2} < 0$ **rel. Max**

g (xe [1]) -> Gewinn $\Rightarrow$

$$\frac{\left(a \cdot \left(\sqrt{b\cdot c^2+2} - \sqrt{b}\cdot c \right) \cdot e^{\frac{-\sqrt{b}\cdot\sqrt{b\cdot c^2+2}\cdot c}{2}} - 2\cdot\sqrt{b}\cdot e^{\frac{b\cdot c^2}{2}+1/2} \cdot d \right) \cdot e^{\frac{-b\cdot c^2}{2} - 1/2}}{2\cdot\sqrt{b}}$$

Mit dem Befehl fMax:

fmax (g (x),x**)** | a>0 $\Rightarrow$ x =∞ **or** x= - ∞ **or** $x = \dfrac{-(\sqrt{b\cdot c^2+2}-\sqrt{b}\cdot c)}{2\cdot\sqrt{b}}$ **or** $x = \dfrac{(\sqrt{b\cdot c^2+2}+\sqrt{b}\cdot c)}{2\cdot\sqrt{b}}$

In dieser Auswahlsendung kommt nur das letzte Resultat als Lösung für unser Problem in Frage. Die zweitletzte Lösung ist negativ und ist schon deshalb keine Lösung für unser Problem.

Beispiel :

10^6->a : 2 · 10^(-4)->b: 39 -> c : 6 · 10^4 -> d
a ***e**^(- b*x^2) -> n (x) $\Rightarrow$ Done
c + d / n (x) -> k (x) $\Rightarrow$ Done
n(x) * (x - k (x)) -> g (x) $\Rightarrow$ Done
zeros (*d* (g (x) , x **)** , x **)** ->xe $\Rightarrow$ [-34.168 73.168}
xe[2] $\Rightarrow$ 73.17
n (xe[2]) $\Rightarrow$ 342 765
g (xe[2]) $\Rightarrow$ 1.16516 E7
g (0) $\Rightarrow$ - 39 060 00

[WINDOW]
xmin =0 xmax =150 xscl = 10
ymin=-10000000 ymax =15000000
[HOME]

graph g (x)



2.5 Taylor-Reihen

ANSATZ: Eine Funktion f soll als Potenzreihe dargestellt werden.

$$f(x) = a_0 + a_1 x + a_2 x^2 + a_3 x^3 + a_4 x^4 + a_5 x^5 + \ldots = \sum_{k=0}^{\infty} a_k x^k$$

Untersuche mit dem CAS die 1. , 2. , ... n-te Ableitung von $f(x) = \sum_{k=0}^{\infty} a_k x^k$ an der Stelle 0

Lösung

Vereinfachung[4]: Wir brechen die Potenzreihe nach dem 20. Glied ab.

Σ (a (k) * xx ^k, k , 0 , 19) -> f (xx)	⇒	Done
d **(** f (x) , x, n**)** | x = 0 **and** n = 1	⇒	a (1)
d **(** f (x) , x, n**)** | x = 0 **and** n = 2	⇒	2 a (2)
d **(** f (x) , x, n**)** | x = 0 **and** n = 3	⇒	6 a (3)
d **(** f (x) , x, n**)** | x = 0 **and** n = 4	⇒	24 a (4)

:
usw.

Vermutung: $f^{(n)}(0) = n! \cdot a(n) = n!\, a_n \quad \Rightarrow \quad a_n = \dfrac{f^{(n)}(0)}{n!}$

FOLGERUNG :
Mit $f(0) = a(0) = a_0 = \dfrac{f(0)}{0!}$ gilt:

$$f(x) = \frac{f(0)}{0!} + \frac{f^{(1)}(0)}{1!}x + \frac{f^{(2)}(0)}{2!}x^2 + \frac{f^{(3)}(0)}{3!}x^3 + \ldots\ldots = \sum_{k=0}^{\infty}\frac{f^{(k)}(0)}{k!}x^k$$

Taylor-Polynom:

CAS - Befehl: **taylor (** f (x) , x , n [, xo] **)**

Beispiele: **taylor (sin (** x **)** , x , 5 **)** ⇒ $\dfrac{x^5}{120} - \dfrac{x^3}{6} + x$

Mit Trick: $f(x) = e^{-\sqrt{x}}$

taylor (***e*** **(-** t **)** , t , 5 **)** | t =√ (x) ⇒ $\dfrac{x^2}{24} - \dfrac{x^{3/2}}{6} + \dfrac{x}{2} - \sqrt{x} + 1$

Problem: **taylor (** x /**sin (**x**)** , x , 4 **)** **Wartezeit ca. 5 min**

[4] Der TI-92 kann für k = 0 ... ∞ die Reihe nicht ableiten.

Literaturverzeichnis

[1] Bachmann Heinz, Einführung in die Analysis 3 Sabe, 1975

[2] Braun M., Differentialgleichungen und ihre Anwendungen, Springer-Verlag Berlin Heidelberg New York, 1979

[3] Burg Klemens / Haf Herbert / Wille Friedrich, Höhere Mathematik für Ingenieure I, B.G Teubner, Stuttgart, 1985

[4] DMK, Analysis, Orell Füssli Verlag Zürich und Wiesbaden, 1989

[5] DMK, Formeln und Tafeln, Orell Füssli Verlag Zürich und Wiesbaden, 1984

[6] Gerthsen Christian, Physik, Springer-Verlag Berlin Heidelberg New York, 1966

[7] Heuser Harro, Gewöhnliche Differentialgleichungen , B.G. Teubner Stuttgar, 1989

[8] Kirchgraber Urs, Differentialgleichungen in die Schule, Skriptum Departement Mathematik ETH Zürich, 1994

[9] Preckur Helmuth, Analysis 3, Mentor Verlag, München, 1988

[10] Storrer Hans Heiner, Einführung in die mathematische Behandlung der Naturwissenschaften, Birkhäuser Verlag, Basel, 1986

[11] Strang Gilbert, Calculus, Wellesley-Cambridge Press, Wellesley, 1991

[12] Vogt Herbert, Grundkurs Mathematik für Biologen, B.G. Teubner, Stuttgart, 1983